


Moonlander



The year 2048. On a parallel earth the moon is explored for the first time. The previous 1023 missions from “Zspace eXploration” failed. Will mission 1024 land on the moon?

; Moonlander, game 49 in 1K hires

? * TORNADO *

```
ORG    #4009                ;#4009
DUMP 49161
```

```
basic    LD    B,5            ; preset for 48K bug
          JR    init0

          DEFB 236,212,28      ; The BASIC
          DEFB 126             ; fully placed over sysvar
          DEFB 143,0,18        ; start to BASIC=#4009
```

```
eline    DEFW last            ; needed by loading
chadd     DEFW last-1
xptr      DEFW 0
stkbot    DEFW last
stkend    DEFW last
berg      DEFB 0
mem        DEFW 0
          DEFB 128
```

```
init1     JP    init          ; init can be anywhere
```

```

; all above reusable AFTER loading

lastk      DEFB 255,255,255      ; used by ZX81
margin     DEFB 55               ; used by ZX81
nxtlin     DEFW basic           ; reusable after load

init0      XOR  A                ; delay intrupts by
fuel       DEFB 254              ; CP n ; skip flagx
flagx      DEFB 0

          EX  AF,AF'             ; intruptcounter reset
          DEFB #3A               ; LD A,(nn) ; skip taddr

taddr      DEFW 3213             ; used by ZX81
          LD  E,L                ; low byte equal 48K bug
          DEFB #3A               ; LD A,(NN) ; skip frames

frames     DEFW 65535            ; used by ZX81
coords     JR  init1             ; useable
prcc       DEFB 188              ; used by ZX81
sposn      DEFB 33,24            ; used by ZX81
cdflag     DEFB 64               ; used by ZX81

ground     DEFB 255,255,255,255,255,255
          DEFB 255,255,255,255,255,255
          DEFB 255,255,255,85,85
          DEFB 85,85,255,255,255
          DEFB 255,255,255,255,255
          DEFB 255,255,255,255,255

hr          LD  HL,lowres+#8000   ; the lowres display
          LD  BC,#311            ; minimum needed
          LD  A,#1E
          LD  I,A
          LD  A,#FB
          CALL #2B5

hr00        LD  B,5               ; sync hires display
          DJNZ hr00

          LD  HL,screen           ; the screen to show
          LD  BC,#40FF           ; B pointer C high
          LD  A,B
          LD  I,A                ; set highbyte HR
          LD  E,5                ; empty pointer for erase
          EXX
          LD  DE,ylander          ; where is the lander?
          LD  B,176              ; 176 lines

nline      LD  A,(DE)            ; get ylander
          SUB  B                  ; test to show
          JR  Z,copyudg          ; if so copy data

erase      EXX
          LD  D,B                ; D now #40
          PUSH DE                 ; save original DE
          XOR  A
          DEC  DE                 ; clear old udg
          LD  (DE),A
          DEC  DE
          LD  (DE),A
          DEC  DE
          LD  (DE),A
          POP  DE                 ; get DE back

```

```

        EXX
        DEC B                ; 1 line less
        JP lbuf+#8000        ; show line

copyudg  EXX
        LD D,B              ; D now #40
xlander  LD E,0              ; E will alter in xlander
        LDI                 ; copy data
        LDI
        LDI
        EXX
        INC E                ; goto next y
        DEC B                ; 1 line less
        JP lbuf+#8000

; end of HR-routine
exit      EXX                ; clear a final udg
        DEC E
        LD (DE),A
        DEC E
        LD (DE),A
        DEC E
        LD (DE),A
        LD A,(HL)           ; timing 2x
        LD A,(HL)

        LD A,ground*256/256
        CALL gndline+#8000   ; now show landingground
        CALL #220
        LD IX,hr
        JP #2A4

gndline   LD R,A
        DEFW 0,0,0,0,0,0,0,0
        DEFW 0,0,0,0,0,0,0,0
        JP #292              ; back from intrupt

ylander   DEFB 255,99,98,97,96,95,94,93,92
        DEFB 91,90,89,88,87,86,85,84
ythruster DEFB 255,255,255    ; motor can be shut off

lbuf      LD R,A
        DEFW 0,0,0,0,0
        DEFW 0,0,0,0,0
        JP NZ,nline          ; 48K bug
        JP exit              ; 48K bug

testland  LD HL,crash         ; preset on crash
        LD A,(xpos+1)
        SUB 67
        CP 15
        JR NC,setMessage    ; not on platform

        LD A,(dypos+1)
        CP 251
        JR C,setMessage     ; too hard landed, crash

        LD HL,hardland       ; hard landing?
        CP 253
        LD A,(fuel)          ; get remaining fuel
        JR C,setMessage
        LD HL,softland
        ADD A,A              ; softlanding double score

```

```

setscore    PUSH HL                ; save text
            LD HL,score
            CALL setvalue          ; set score
            POP HL

setmsg      LD DE,scrmsg
            LD A,(HL)              ; get lander position
            INC HL
            LD BC,6
            LDIR
            LD HL,ylander          ; set lander after landing
            LD B,17

sety2       LD (HL),A
            INC HL
            DEC A
            DJNZ sety2
            XOR A
            LD (msgscr),A          ; allow line2 in lowres

            LD HL,score-1          ; test hiscore
            LD DE,hiscore-1
            LD C,4
            INC HL
            INC DE
            DEC C
            JR Z,start
            LD A,(DE)
            CP (HL)
            JR Z,fihi
            CALL C,#19F9

fihi        LD A,(lastk)           ; game over, wait for
            SUB %10111111          ; newline
            JR NZ,start

            LD (dypos+1),A         ; clear any dy
            LD (dxpos+1),A         ; clear any dx

            LD A,118               ; no line2 in lowres
            LD (msgscr),A

            LD A,125
            LD (fuel),A            ; set fuel

            LD A,176               ; start of lander
            LD (ypos+1),A

rseed       LD HL,(frames)
            LD DE,0
            ADD HL,DE
            INC HL
            LD A,H
            AND #1F
            LD H,A
            LD (rseed+1),HL
            LD A,(HL)

findx       LD (xpos+1),A          ; a random entry
            SUB 80
            JR NC,findx

disploop    XOR A                  ; skip display
            LD (ylander),A

            LD HL,udglander        ; the base lander
            LD DE,screen            ; copy it to screen

```

```

baseudg      LD    BC,#15FF
             LDI
             LDI
             LDI
             DJNZ baseudg          ; copy full lander

             LD    A,(fuel)
             OR    A
             JR    Z,fuel0        ; no move possible

; keys for left right
             LD    A,%11011111    ; read OP
             IN    A,(254)
             LD    DE,#010C      ; preset move right
             LD    HL,screen
             RRA
             JR    NC,move
             LD    DE,#FF30      ; preset move left
             INC    HL
             INC    HL
             RRA
             JR    NC,move
fuel0         LD    DE,0          ; no move
move          LD    B,3          ; 3 puffs to show
setpush      LD    A,(HL)        ; get top of udg
             OR    E             ; add side puff
             LD    (HL),A
             INC    HL           ; ld a,6
             INC    HL           ; add a,1
             INC    HL           ; ld l,a
             INC    HL           ; when not over 256
             INC    HL
             DJNZ setpush        ; set all puffs
             LD    A,E
             OR    A
             JR    Z,skipdec

             LD    HL,fuel        ; we used fuel
             DEC    (HL)

skipdec      LD    A,D           ; get direction

dxpos        ADD    A,0          ; add old movement
             LD    (dxpos+1),A    ; set new direction

; 0-3 -sh 4 +sh 5-6-7
xpos         LD    C,0           ; get xpos
             ADD    A,C          ; add dx

             CP    145           ; test out of screen
             JR    C,dodx

dodx         LD    A,C           ; undo move
             LD    (xpos+1),A     ; set new xpos

             PUSH    AF          ; save xpos
             AND    7            ; pixelpos only
             SUB    4            ; 4 is base
             LD    C,A          ; save result
             JR    Z,noshift     ; base udg is ok
             JR    NC,rrshift    ; shift right

rlshift      LD    HL,screen+62  ; shift left

```

```

rlshft      LD    B,21
            RL    (HL)
            DEC   HL
            RL    (HL)
            DEC   HL
            RL    (HL)
            DEC   HL
            DJNZ  rlshft
            INC   C
            JR    NZ,rlshift
            JR    noshift

rrshift     LD    HL,screen
            LD    B,20
rrshft      RR    (HL)
            INC   HL
            RR    (HL)
            INC   HL
            RR    (HL)
            INC   HL
            DJNZ  rrshft
            DEC   C
            JR    NZ,rrshift

noshift     POP   AF                      ; get xpos
            AND   #F8                      ; bytes only
            RRCA
            RRCA
            RRCA
            LD    (xlander+1),A           ; set bytepos

            LD    HL,showdy+1
            LD    (HL),20                  ; default show puffs
            LD    A,(fuel)
            LD    E,5
            SUB   E
            JR    C,toolow                ; not enough fuel for puff

            LD    A,%11111011             ; key for up
            IN    A,(254)
            RRA
            LD    A,5
            JR    NC,decfuel              ; we will reduce speed
toolow      LD    (HL),17                  ; no puff to show

            XOR   A
            LD    E,A
            LD    (ythrust),A             ; do not show power up

decfuel     LD    HL,fuel
            LD    E,A
            LD    A,(HL)
            SUB   E
            LD    (HL),A

; display fuel on screen
            LD    HL,fuelvis
            CALL  setvalue

            LD    A,E                      ; get powerup, 0 or 5

dypos       ADD   A,0                      ; old speed
            DEC   A                        ; gravity
            LD    (dypos+1),A              ; set new speed

```

```

        LD     E,A

ypos    LD     C,0                ; ypos
        ADD    A,C
        CP     176                ; out of screen
        LD     HL,orbit           ; means in orbit
        JP     NC,setmsg
        LD     (ypos+1),A
        SUB    18
        CP     230
        JP     NC,testland        ; we contacted the moon

        LD     A,E
        LD     D,1
        CP     100

        JR     C,pos
        XOR    A
        SUB    E
        LD     D,255
pos      LD     E,A

        LD     A,18
        SUB    E
        LD     (endwait+1),A      ; constant timing

        LD     A,E
        OR     A
        JR     Z,ymoveok

showdy   LD     B,17              ; we move nr of pixels down
        LD     HL,ylander
        LD     A,C
        ADD    A,D
        LD     C,A
sety     LD     (HL),A            ; set ypos per line
        INC    HL
        DEC    A
        DJNZ   sety

        LD     HL,frames
        LD     A,(HL)
wfr      CP     (HL)
        JR     Z,wfr
        DEC    E
        JR     NZ,showdy

        LD     A,(HL)
endwait  SUB    0                ; remaining time no move down
wfr2     CP     (HL)
        JR     NZ,wfr2

ymoveok  JP     disploop

setvalue LD     BC,#264          ; 2 fields, 100
set0     LD     (HL),27
set1     INC    (HL)
        SUB    C                ; sub 100 or 10
        JR     NC,set1
        INC    HL
        ADD    A,C
        LD     C,10
        DJNZ   set0
        ADD    A,28

```

```

        LD      (HL),A
        RET

n        EQU    27
x        EQU    101

softland DEFB 17,"S"-n,"O"-n,"F"-n,"T"-n,118
hardland DEFB 16,"H"-n,"A"-n,"R"-n,"D"-n,118
crash    DEFB 8,"C"-n,"R"-n,"A"-n,"S"-n,"H"-n

lowres   DEFB 118
fuelvis  DEFB 28,28,28,0,0,0,0,0

        DEFB "M"+x,"O"+x,"O"+x,"N"+x,"L"+x,"A"+x
        DEFB "N"+x,"D"+x,"E"+x,"R"+x,0,0,0,0

score    DEFB 28,28,28,0
hiscore  DEFB 28,28,28
        DEFB 118
msgscr   DEFB 118,0,0,0,0,0
        DEFB "L"-n,"A"-n,"N"-n,"D"-n,"I"-n,"N"-n,"G"-n,14
scrmsg   DEFB 118
        DEFS 5

udglander DEFB 0,255,0
        DEFB 1,255,128
        DEFB 1,26,128
        DEFB 1,218,128
        DEFB 1,189,128
        DEFB 1,122,128
        DEFB 3,26,192
        DEFB 7,255,224
        DEFB 5,178,160
        DEFB 5,90,160
        DEFB 5,86,32
        DEFB 5,179,160
        DEFB 7,255,224
        DEFB 7,255,224
        DEFB 15,255,240
        DEFB 13,231,176
        DEFB 8,195,16
        DEFB 0,0,0
        DEFB 5,36,160
        DEFB 5,36,160

orbit    DEFB 0
        DEFB "O"-n,"R"-n,"B"-n,"I"-n,"T"-n,118

; overwritten with the shifted udg
screen   EQU    $
init     LD      IX,hr           ; 04 Hires mode
        LD      SP,#4400        ; 07
        LD      H,#3F           ; 09 #3fxx
        LD      D,#BF           ; 11 #bfxx
        LDIR     ; 13 repair 48K bug
        LD      HL,#4016        ; 16
cldisp   DEC     L               ; 17
        LD      (HL),B          ; 18
        JR      NZ,cldisp
        JP      start

vars     DEFB 128
last     EQU    $

```